

Haskell

Starting Out

<http://igm.univ-mlv.fr/~vialette/?section=teaching>

Stéphane Vialette

LIGM, Université Paris-Est Marne-la-Vallée

November 11, 2017



Ready, set, go!



Ready, set, go!

ghc's interactive mode

```
barbalala: ghci
GHCi, version 7.10.3: http://www.haskell.org/ghc/
:? for help
ghci> :set prompt "ghci> "
ghci> 1 + 2
3
ghci> 3 * 4
12
ghci> 5 - 6
-1
ghci> 7 / 8
0.875
ghci>
```



Ready, set, go!

ghc's interactive mode

```
ghci> True && True
True
ghci> False || True
True
ghci> not False
True
ghci> not (True && True)
False
```



Ready, set, go!

ghc's interactive mode

```
ghci> 1 == 1
True
ghci> 1 == 2
False
ghci> 1 /= 2
True
ghci> 1 /= 1
False
ghci> False && True
False
```



Ready, set, go!

ghc's interactive mode

```
ghci> 1 == True
<interactive>:17:1:
    No instance for (Num Bool) arising from the literal '1'
    In the first argument of '(==)', namely '1'
    In the expression: 1 == True
    In an equation for 'it': it = 1 == True
ghci>
```



Functions

ghc's interactive mode

* is a function that takes two numbers and multiplies them.

As you've seen, we call it by sandwiching it between them. This is what we call an **infix function**.

Most functions that aren't used with numbers are prefix functions.



Calling functions



Suppose that `f` is a function that takes two integers as arguments and returns some integer (in haskell `f :: Int -> Int -> Int`).

Call `f` **without parentheses**:

```
f 3 4
```

Do not write:

```
f(3, 4)
```



Functions

ghc's interactive mode

```
ghci> succ 1
2
ghci> :type succ
succ :: Enum a => a -> a
ghci> :type succ 1
succ 1 :: (Enum a, Num a) => a
ghci>
ghci> succ 1.2
2.2
ghci> :type succ 1.2
succ 1.2 :: (Enum a, Fractional a) => a
ghci> succ False
True
ghci> :t succ False
succ False :: Bool
ghci> succ True
*** Exception: ghci.Enum.Bool.succ: bad argument
```



Functions

ghc's interactive mode

```
ghci> min 2 1
1
ghci> :type min
min :: Ord a => a -> a -> a
ghci> :type min 2 1
min 2 1 :: (Num a, Ord a) => a
ghci>
ghci> max 2 1
2
ghci> :type max
max :: Ord a => a -> a -> a
ghci> :type max 2 1
max 2 1 :: (Num a, Ord a) => a
```



Functions

ghc's interactive mode

Function application has the highest precedence of them all. What that means for us is that these two statements are equivalent.

```
ghci> succ 9 + max 5 4 + 1
16
ghci> (succ 9) + (max 5 4) + 1
16
ghci>
```



Baby's first functions

```
ghci> let doubleMe x = x + x
ghci> :type doubleMe
doubleMe :: Num a => a -> a
ghci> doubleMe 3
6
ghci> doubleMe 3.0
6.0
ghci> doubleMe 9/3
6.0
ghci> :type (/)
(/) :: Fractional a => a -> a -> a
ghci>
```



Baby's first functions

```
ghci> let doubleUs x y = 2*x + 2*y
ghci> :type doubleUs
doubleUs :: Num a => a -> a -> a
ghci> doubleUs 4 9
26
ghci> doubleUs 2.3 34.2
73.0
ghci> doubleUs 28 88 + doubleMe 123
478
ghci> doubleUs doubleMe 2 doubleMe 10
<interactive>:11:1:
    No instance for (Num ((a0 -> a0) -> a0 -> a0))
      arising from a use of 'it'
    In a stmt of an interactive GHCi command: print it
ghci> doubleUs (doubleMe 2) (doubleMe 10)
48
ghci>
```



Baby's first functions

```
doubleMe x = x + x
```

```
doubleUs x y = doubleMe x + doubleMe y
```

This is a very simple example of a common pattern you will see throughout Haskell. Making basic functions that are obviously correct and then combining them into more complex functions



Baby's first functions

Write a function that multiplies a number by 2 but only if that number is smaller than or equal to 100 because numbers bigger than 100 are big enough as it is!

```
doubleSmallNumber x = if x > 100
                        then x
                        else 2*x
```

The difference between Haskell's `if` statement and `if` statements in imperative languages is that the `else` part is mandatory in Haskell.



Baby's first functions

Another thing about the `if` statement in Haskell is that it is an expression.

An expression is basically a piece of code that returns a value.

5 is an expression because it returns 5, `1+2` is an expression, `x+y` is an expression because it returns the sum of `x` and `y`.

Because the `else` is mandatory, an `if` statement will always return something and that's why it's an expression.

If we wanted to add one to every number that's produced in our previous function, we could have written its body like this:

```
doubleSmallNumber' x = (if x > 100 then x else 2*x) + 1
```



Baby's first functions

Note the `'` at the end of the function name.

That apostrophe doesn't have any special meaning in Haskell's syntax. It's a valid character to use in a function name.

We usually use `'` to either denote a strict version of a function (one that isn't lazy) or a slightly modified version of a function or a variable.

Because `'` is a valid character in functions, we can make a function like this:

```
conanO'Brien = "It's a-me, Conan O'Brien!"
```



Baby's first functions

```
doubleSmallNumber x = if x > 100 then x else 2*x
```

```
doubleSmallNumber' x  
  | x < 100    = x  
  | otherwise = 2*x
```



Functional thinking



An introduction to lists

```
ghci> let lostNumbers = [4,8,15,16,23,42]
ghci> lostNumbers
[4,8,15,16,23,42]
ghci> :type lostNumbers
lostNumbers :: Num t => [t]
ghci>
```

```
ghci> [1,2,3,4] ++ [9,10,11,12]
[1,2,3,4,9,10,11,12]
ghci> "hello" ++ " " ++ "world"
"hello world"
ghci> :type "toto"
"toto" :: [Char]
ghci> ['w','o'] ++ ['o','t']
"woot"
ghci>
```



An introduction to lists

When you put together two lists (even if you append a singleton list to a list, for instance: `[1,2,3] ++ [4]`), internally, Haskell has to walk through the whole list on the left side of `++`.

That's not a problem when dealing with lists that aren't too big.

However, putting something at the beginning of a list using the `:` operator (also called the cons operator) is instantaneous:

```
ghci> 'A' : " SMALL CAT"
"A SMALL CAT"
ghci> 5 : [1,2,3,4,5]
[5,1,2,3,4,5]
ghci>
```

Notice how `:` takes a number and a list of numbers or a character and a list of characters, whereas `++` takes two lists.



An introduction to lists

If you want to get an element out of a list by index, use `!!`. The indices start at 0.

```
ghci> [1,2,3,4,5] !! 0
1
ghci> "Hello" !! 1
'e'
ghci> [1,2,3,4,5] !! (-1)
*** Exception: ghci.(!): negative index
ghci> [1,2,3,4,5] !! 5
*** Exception: ghci.(!): index too large
ghci>
```

You will rarely, if ever, use `!!`. If you happen to use `!!`, you should think again.



An introduction to lists

Lists can also contain lists. They can also contain lists that contain lists that contain lists ...

```
ghci> let l = [[1,2,3],[4,5,6],[7,8,9]]
ghci> l
[[1,2,3],[4,5,6],[7,8,9]]
ghci> :type l
l :: Num t => [[t]]
ghci> l ++ [[6,6,6]]
[[1,2,3],[4,5,6],[7,8,9],[6,6,6]]
ghci> [0,0,0]:l
[[0,0,0],[1,2,3],[4,5,6],[7,8,9]]
ghci> l !! 1
[4,5,6]
ghci>
```



An introduction to lists

```
ghci> 1 : 2 : 3 : 4 : []  
[1,2,3,4]  
ghci> 1 : 2 : 3 : [4]  
[1,2,3,4]  
ghci> 1 : 2 : [3, 4]  
[1,2,3,4]  
ghci> 1 : [2,3, 4]  
[1,2,3,4]  
ghci> [1,2,3, 4]  
[1,2,3,4]
```



An introduction to lists

```
ghci> let l1 = [1,2,3]
ghci> let l2 = [4,5,6]
ghci> l1 : l2
<interactive>:36:1:
    Non type-variable argument in the constraint: Num [t]
    ...
ghci> l1 ++ l2
[1,2,3,4,5,6]
```



An introduction to lists

```
ghci> let l = 1 : 3 : 1
ghci> l !! 0
1
ghci> l !! 1
3
ghci> l !! 2
1
ghci> l !! 3
3
ghci> l -- don't do this
[1,2,1,2,1,2,1,2,1,2,1,2,1,2,1,2,1,...
```



An introduction to lists

Lists can be compared if the stuff they contain can be compared.

When using `<`, `<=`, `>` and `>=` to compare lists, they are compared in lexicographical order. First the heads are compared. If they are equal then the second elements are compared, etc.

```
ghci> [3,2,1] > [2,1,0]
True
ghci> [3,2,1] > [2,10,100]
True
ghci> [3,4,2] > [3,4]
True
ghci> [3,4,2] > [2,4]
True
ghci> [3,4,2] == [3,4,2]
True
```



An introduction to lists

Some basic functions that operate on lists

`head` takes a list and returns its head. The head of a list is basically its first element.

```
ghci> head "Hello"  
'H'  
ghci> head [1,2,3,4,5]  
1  
ghci> head []  
*** Exception: ghci.head: empty list
```



An introduction to lists

Some basic functions that operate on lists

`tail` takes a list and returns its tail. In other words, it chops off a list's head.

```
ghci> tail "Hello"
"ello"
ghci> tail [1,2,3,4,5]
[2,3,4,5]
ghci> tail [1]
[]
ghci> tail ['a']
""
ghci> tail []
*** Exception: ghci.tail: empty list
```



An introduction to lists

Some basic functions that operate on lists

`last` takes a list and returns its last element.

```
ghci> last "Hello"  
'o'  
ghci> last [1,2,3,4,5]  
5  
ghci> last []  
*** Exception: ghci.last: empty list
```



An introduction to lists


Some basic functions that operate on lists

`init` takes a list and returns everything except its last element.

```
ghci> init "Hello"  
"Hell"  
ghci> init [1,2,3,4,5]  
[1,2,3,4]  
ghci> init [1]  
[]  
ghci> init []  
*** Exception: ghci.init: empty list
```



An introduction to lists



An introduction to lists

Some basic functions that operate on lists

`length` takes a list and returns its length, obviously.

```
ghci> length "Hello"  
5  
ghci> length [1,2,3,4,5]  
5  
ghci> length []  
0
```



An introduction to lists

Some basic functions that operate on lists

`null` checks if a list is empty. If it is, it returns `True`, otherwise it returns `False`.

Use this function instead of `xs == []` (if you have a list called `xs`)

```
ghci> null "hello"  
False  
ghci> null [1,2,3,4,5]  
False  
ghci> null []  
True
```



An introduction to lists

Some basic functions that operate on lists

`reverse` reverses a list.

```
ghci> reverse "hello"  
"olleh"  
ghci> reverse [1,2,3,4,5]  
[5,4,3,2,1]  
ghci> reverse []  
[]  
ghci> "abcd" == reverse (reverse "abcd")  
True
```



An introduction to lists

Some basic functions that operate on lists

`take` takes a number and a list. It extracts that many elements from the beginning of the list.

```
ghci> :type take
take :: Int -> [a] -> [a]
ghci> take 0 [1,2]
[]
ghci> take 1 [1,2]
[1]
ghci> take 2 [1,2]
[1,2]
ghci> take 3 [1,2]
[1,2]
ghci> take 0 []
[]
ghci> take 1 []
[]
```



An introduction to lists

Some basic functions that operate on lists

`drop` works in a similar way, only it drops the number of elements from the beginning of a list.

```
ghci> :type drop
drop :: Int -> [a] -> [a]
ghci> drop 0 [1,2,3]
[1,2,3]
ghci> drop 1 [1,2,3]
[2,3]
ghci> drop 2 [1,2,3]
[3]
ghci> drop 3 [1,2,3]
[]
ghci> drop 4 [1,2,3]
[]
```



An introduction to lists

Some basic functions that operate on lists

`maximum` takes a list of stuff that can be put in some kind of order and returns the biggest element. `minimum` returns the smallest.

```
ghci> :type minimum
minimum :: (Ord a, Foldable t) => t a -> a
ghci> minimum [3,4,2,5,1,6,9,8,7]
1
ghci> :type maximum
maximum :: (Ord a, Foldable t) => t a -> a
ghci> maximum [3,4,2,5,1,6,9,8,7]
9
ghci> minimum []
*** Exception: ghci.minimum: empty list
ghci> maximum []
*** Exception: ghci.maximum: empty list
```



An introduction to lists

Some basic functions that operate on lists

`sum` takes a list of numbers and returns their sum.

`product` takes a list of numbers and returns their product.

```
ghci> :type sum
sum :: (Num a, Foldable t) => t a -> a
ghci> sum []
0
ghci> sum [1,2,3,4,5]
15
ghci> :type product
product :: (Num a, Foldable t) => t a -> a
ghci> product []
1
ghci> product [1,2,3,4,5]
120
ghci> let fact n = product [1..n]
ghci> fact 5
120
```



An introduction to lists

Some basic functions that operate on lists


`elem` takes a thing and a list of things and tells us if that thing is an element of the list.

It's usually called as an infix function because it's easier to read that way.

```
ghci> :type elem
elem :: (Eq a, Foldable t) => a -> t a -> Bool
ghci> 3 `elem` [2,1,3,5,4]
True
ghci> elem 3 [2,1,3,5,4]
True
ghci> 6 `elem` [2,1,3,5,4]
False
ghci> elem 6 [2,1,3,5,4]
False
ghci>
```



Texas ranges



Texas ranges

```
ghci> [1,2,3,4,5,6,7,8,9,10]
[1,2,3,4,5,6,7,8,9,10]
ghci> [1..10]
[1,2,3,4,5,6,7,8,9,10]
ghci> [10..1]
[]
ghci> [1.0..10.0] -- don't do this!
[1.0,2.0,3.0,4.0,5.0,6.0,7.0,8.0,9.0,10.0]
ghci> ['a'..'z']
"abcdefghijklmnopqrstuvwxyz"
ghci> ['A'..'Z']
"ABCDEFGHIJKLMNOPQRSTUVWXYZ"
ghci>
```



Texas ranges

Ranges are cool because you can also specify a step.

```
ghci> [10,13..20]
[10,13,16,19]
ghci> ['a','e'..'z']
"aeimquy"
ghci> [1,2,4,8,16..100] -- expecting the powers of 2 !
<interactive>:181:12: parse error on input '['. '
```



Texas ranges

Ranges are cool because you can also specify a step.


```
ghci> [10,13..20]
[10,13,16,19]
ghci> ['a','e'..'z']
"aeimquy"
ghci> [1,2,4,8,16..100] -- expecting the powers of 2 !
<interactive>:181:12: parse error on input '['. '
```



Texas ranges

Do not use floating point numbers in ranges!

```
ghci> [0.1, 0.3 .. 1]
[0.1,0.3,0.5,0.7,0.8999999999999999,1.0999999999999999]
ghci> [1, 0.8 .. 0]
[1.0,0.8,0.6000000000000001,0.40000000000000013,
 0.20000000000000018,2.220446049250313e-16]
ghci>
```



Texas ranges

You can also use ranges to make infinite lists by just not specifying an upper limit.

Because Haskell is lazy, it won't try to evaluate the infinite list immediately.

```
ghci> let l = [1..]
ghci> :type l
l :: (Num t, Enum t) => [t]
ghci> take 10 l
[1,2,3,4,5,6,7,8,9,10]
ghci> l -- don't do this
[1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20,
21,22,23,24,25,26,27,28,29,30,31,32,33,34,35,36,37,
38,39,40,41,42,43,44,45,46,47,48,49,50,51,52,53,54,
55,56,57,58,59,60,61,62,63,...]
```



I'm a list comprehension



I'm a list comprehension

A basic comprehension for a set that contains the first ten even natural numbers is

$$\{2x \mid x \in \mathbb{N}, x \leq 10\}$$

In Haskell

```
ghci> [2*x | x <- [1..10]]  
[2,4,6,8,10,12,14,16,18,20]  
ghci>
```



I'm a list comprehension

```
ghci> [x*2 | x <- [1..10], x*2 >= 12]
[12,14,16,18,20]
ghci> [ x | x <- [50..100], x `mod` 7 == 3]
[52,59,66,73,80,87,94]
ghci> :{  -- starts multiline-input mode
ghci| let ab xs = [if x<10 then "a" else "b" |
ghci|                x <- xs, odd x]
ghci| :}  -- terminates multiline-input mode
ghci> ab [7..13]
["a","a","b","b"]
ghci> [ x | x <- [10..20], x /= 13, x /= 15, x /= 19]
[10,11,12,14,16,17,18,20]
ghci>
```



I'm a list comprehension

```
ghci> [x+y | x <- [1,2,3], y <- [100,200,300]]
[101,201,301,102,202,302,103,203,303]
ghci> :{
ghci| [x+y | x <- [1,2,3], y <- [100,200,300],
ghci|      x+y > 200]
ghci| :}
[201,301,202,302,203,303]
ghci> :{
ghci| [x+y | x <- [1,2,3], y <- [100,200,300],
ghci|      x+y > 200, x+y < 300]
ghci| :}
[201,202,203]
ghci>
```



I'm a list comprehension

```
ghci> let length' xs = sum [1 | _ <- xs]
ghci> length' []
0
ghci> length' [1..100]
100
ghci> :{
ghci| let removeNonUppercase cs = [c |
ghci|                               c <- cs,
ghci|                               c `elem` ['A'..'Z']]
ghci| :}
ghci> removeNonUppercase "Hahaha! Ahahaha!"
"HA"
ghci> removeNonUppercase "IdontLIKEFROGS"
"ILIKEFROGS"
ghci>
```



I'm a list comprehension

Nested list comprehensions are also possible if you're operating on lists that contain lists.

```
ghci> let xss = [[1,2,3],[4,5],[6,7,8,9,10]]
ghci> [[x | x <- xs, even x] | xs <- xss]
[[2],[4],[6,8,10]]
ghci> [[x | x <- xs, even x] | xs <- xss, length' xs > 2]
[[2],[6,8,10]]
ghci>
```



I'm a list comprehension

```
ghci> let fibs = 0 : 1 : [ a + b | (a, b) <- zip fibs (tail fibs) ]
ghci> :t fibs
fibs :: Num b => [b]
ghci> take 0 fibs
[]
ghci> take 10 fibs
[0,1,1,2,3,5,8,13,21,34]
```



I'm a list comprehension

```
ghci> let binaries = [b : bs | bs <- "" : binaries, b <- ['0','1']]
ghci> take 2 binaries
["0","1"]
ghci> take 6 binaries
["0","1","00","10","01","11"]
ghci> take 14 binaries
["0","1","00","10","01","11","000","100","010","110","001","101","011","100"]
ghci> take 6 (filter (\bs -> last bs == '0') binaries)
["0","00","10","000","100","010"]
```



Tuples



Tuples

In some ways, tuples are like lists – they are a way to store several values into a single value.

However, there are a few fundamental differences. A list of numbers is a list of numbers. That's its type and it doesn't matter if it has only one number in it or an infinite amount of numbers. Tuples, however, are used when you know exactly how many values you want to combine and its type depends on how many components it has and the types of the components.

They are denoted with parentheses and their components are separated by commas.

Another key difference is that they don't have to be homogenous. Unlike a list, a tuple can contain a combination of several types.



Tuples

```
ghci> :type (1,2)
(1,2) :: (Num t1, Num t) => (t, t1)
ghci> :type (1,2,3)
(1,2,3) :: (Num t2, Num t1, Num t) => (t, t1, t2)
ghci> [(1,2),(8,11),(4,5)]
[(1,2),(8,11),(4,5)]
ghci> [(1,2),(8,11,5),(4,5)]
<interactive>:68:8:
    Couldn't match expected type ...
```



Tuples

```
ghci> :type ('a', 1, "hello")
('a', 1, "hello") :: Num t => (Char, t, [Char])
ghci> (1, 2) < (3, 4)
True
ghci> (1, 2) < (0, 1)
False
ghci> (1, 2, 3) < (1, 2)
<interactive>:74:13:
    Couldn't match expected type ...
```



Tuples

Two useful functions that operate on pairs

```
ghci> fst ('a', 2)
'a'
ghci> snd ('a', 2)
2
ghci> fst ('a', 2, "hello")
<interactive>:80:5:
    Couldn't match expected type ...
```



Tuples

Two useful functions that operate on pairs

```
ghci> :type fst
fst :: (a, b) -> a
ghci> :type snd
snd :: (a, b) -> b
ghci> fst (('a', 'b'), ('c', 'd'))
('a', 'b')
ghci> fst (('a', 'b', 'c'), ('d', 'e', 'f'))
('a', 'b', 'c')
ghci> fst (('a', 'b', 'c'), ('d', 'e', 'f', 'g'))
('a', 'b', 'c')
```



Tuples

zip



Tuples

zip

```
ghci> :t zip
zip :: [a] -> [b] -> [(a, b)]
```

```
ghci> zip [1,2,3,4] ['a','b','c','d']
[(1,'a'),(2,'b'),(3,'c'),(4,'d')]
ghci> zip [1,2,3,4,5] ['a','b','c','d']
[(1,'a'),(2,'b'),(3,'c'),(4,'d')]
ghci> zip [1,2,3,4] ['a','b','c','d','e']
[(1,'a'),(2,'b'),(3,'c'),(4,'d')]
ghci> zip [1,2,3,4] ['a'..]
[(1,'a'),(2,'b'),(3,'c'),(4,'d')]
ghci> zip [1..] ['a','b','c','d']
[(1,'a'),(2,'b'),(3,'c'),(4,'d')]
ghci> zip [1..] ["apple", "orange", "cherry", "mango"]
[(1,"apple"),(2,"orange"),(3,"cherry"),(4,"mango")]
```



Tuples


zipWith

```
ghci> :t zipWith
zipWith :: (a -> b -> c) -> [a] -> [b] -> [c]
ghci> zipWith (+) [0..5] [1,1..]
[1,2,3,4,5,6]
ghci> let rfibs = 0 : 1 : zipWith (+) rfibs (tail rfibs)
ghci> take 10 rfibs
[0,1,1,2,3,5,8,13,21,34]
```



Right triangle

Which right triangle that has integers for all sides and all sides equal to or smaller than 10 has a perimeter of 24?



$$a^2 + b^2 = c^2$$



Right triangle

```
ghci> :{
ghci| let triangles = [(a,b,c) |
ghci|                     a <- [1..10], b <- [1..10],
ghci|                     c <- [1..10]]
ghci| :}
ghci> :{
ghci| let rightTriangles = [(a,b,c) |
ghci|                        (a,b,c) <- triangles,
ghci|                        a^2 + b^2 == c^2]
ghci| :}
ghci> :{
ghci| let rightTriangles' = [(a,b,c) |
ghci|                        (a,b,c) <- triangles,
ghci|                        a^2 + b^2 == c^2,
ghci|                        a+b+c == 24]
ghci| :}
ghci> rightTriangles'
[(6,8,10),(8,6,10)]
ghci>
```



Done!

ABSTRACTION
NO SIDE EFFECTS **LAZY**
STRONG-TYPED
MINIMISES BUGS
ELEGANCE

